/**
 * @(#) BizTripCodesProxy.as
 *
 * @date		2009. 04. 6.
 * @version		1.0
 * @author		Shin, Yunwoo
 * @description	 국경일 및 휴일정보 Proxy - Proxy
 * 				월별 국경일 및 휴일 정보를 가져온다.
 */
package com.skt.common_hr.biztrip_sac.model
{
	import com.skt.common_hr.biztrip_sac.BizTripFacade;
	import com.skt.common_hr.biztrip_sac.model.vo.BizTripCodesVO;
	import com.skt.common_hr.common.Variables;
	import com.skt.common_hr.common.puremvc.CommonProxy;
	import com.wcl.dataloader.*;

	import mx.collections.ArrayCollection;
	import mx.managers.CursorManager;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;

	public class BizTripCodesProxy extends CommonProxy
	{
		public static const NAME:String = "BizTripCodesProxy";

		public var bizTripTypes:ArrayCollection;
		public var countries:ArrayCollection;
		public var costTypes:ArrayCollection;
		public var evidenceTypes:ArrayCollection;
		public var currencyTypes:ArrayCollection;

		/**
		 * 생성자
		 *
		 * @param
		 * @return void
		 */
		public function BizTripCodesProxy(data:Object=null)
		{
			super(NAME, data);
		}


		/**
		 * data load를 수행
		 * 		- DataLoader 를 생성하고 Import Parameter 객체를 생성, RFC를 호출
		 *
		 * @param pernr 사번,대상월
		 * @return void
		 */
		public function load():void
		{
			var param:Object = new Object();
			loadData("Z_HR_T007", param, result, fault);
		}



		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Event Listeners
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 *
		 * @param
		 * @return void
		 */
		public function result(event:ResultEvent):void
		{
			if(event.result["RETCODE"]==0) {
				bizTripTypes 	= parseBizTripCodes(event.result.T_TRTYPE);
				countries 		= parseBizTripCodes(event.result.T_LAND);
				costTypes 		= parseCostTypes(event.result.T_EXPNS);
				evidenceTypes 	= parseBizTripCodes(event.result.T_RECGBN);
				currencyTypes 	= parseBizTripCodes(event.result.T_CURKY);
				facade.sendNotification(BizTripFacade.LOAD_BIZTRIP_CODES_COMPLETE, this);
			}
			else {
				//trace( "failed to 출장 코드 리스트: (" +event.result["RETCODE"] + ") "+event.result["RETTEXT"]);
				facade.sendNotification(BizTripFacade.LOAD_BIZTRIP_CODES_FAILED, this);
			}
		}

		/**
		 * 서버에서 오류가 발생했을 처리
		 *
		 * @param
		 * @return void
		 */
		private function fault(event:FaultEvent):void
		{
			//trace("Fault message in 출장 코드 리스트: " + event.fault.faultString);
			facade.sendNotification(BizTripFacade.LOAD_BIZTRIP_CODES_FAILED, this);
		}





		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Data Parsing functions
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 *
		 * @param
		 * @return 제증명서 종류 Code list
		 */
		private function parseBizTripCodes(srcArr:ArrayCollection):ArrayCollection
		{
			var list:ArrayCollection = new ArrayCollection;
			for each(var rec:Object in srcArr)
			{
				var vo:BizTripCodesVO = new BizTripCodesVO();
				vo.CODE		= rec["CODE"];
				vo.TEXT		= rec["TEXT"];
				//trace(vo.CODE + ", " + vo.TEXT);
				list.addItem(vo);
			}
			return list;
		}

		private function parseCostTypes(srcArr:ArrayCollection):ArrayCollection
		{
			var list:ArrayCollection = new ArrayCollection;

			for each(var rec:Object in srcArr)
			{
				var vo:BizTripCodesVO = new BizTripCodesVO();
				vo.TYPE		= rec["TRTYPE"];
				vo.CODE		= rec["COSTGBN"];
				vo.TEXT		= rec["COSTEXT"];
				list.addItem(vo);
			}
			return list;
		}

	}
}